L'induction mathématique

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en septembre 2004. Le but de ce document est de montrer à l'élève la manière de s'y prendre pour réaliser un développement Maple pour faire une démonstration par induction mathématique. Aussi, dans ce document, l'élève est initié à la structure de boucle informatique.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
```

De la nécessité de faire la preuve

Notre intuition est indispensable pour nous aider à comprendre le monde qui nous entoure. C'est notre intuition qui nous anime dans la recherche de la vérité. Dans la démarche scientifique, elle est le point de départ à la formulation d'une hypothèse que le scientifique mettra à contribution dans l'élaboration d'une théorie visant à donner la meilleure modélisation (explication) du moment. Par le suite, l'expérience confirmera la justesse de la théorie... parfois pour un certain temps.

Avec la démarche mathématique, notre intuition nous anime également dans la recherche de la vérité. Mais, contrairement à la démarche scientifique, elle est le point de départ d'une formule mathématique, d'un théorème qu'il faut prouver par la suite. C'est seulement après la **preuve** de son énoncé qu'on pourra affirmer, *une fois pour toute*, sa véracité. Aucune autre théorie ne pourra par la suite remettre en question sa véracité.

Mettons à l'oeuvre notre intuition dans les deux exemples suivants.

Premier exemple :
Affichons la liste des 20 premiers nombres impairs consécutifs. Nommons cette suite *Impair*.

```
> Impair:=2*k-1;
  Liste:=[seq(eval(Impair,k=n),n=1..20)];
```

$$Impair := 2k - 1$$

$$Liste := [1, 3, 5, 7, 9, 11, 13, 15, 17, 19, 21, 23, 25, 27, 29, 31, 33, 35, 37, 39] \quad \textbf{(2.1)}$$

À partir de cette liste, générons une nouvelle liste que nous appellerons *Somme* dont les éléments sont respectivement le premier terme, la somme des deux premiers termes, la somme des trois premiers, la somme des quatre premiers et ainsi de suite jusqu'à la somme de ces 20 premiers termes de la liste *Impair*.

```
> Somme:=[seq(add(eval(Impair,k=k),k=1..n),n=1..20)];
```

$$Somme := [1, 4, 9, 16, 25, 36, 49, 64, 81, 100, 121, 144, 169, 196, 225, 256, 289, 324, 361, 400] \quad \textbf{(2.2)}$$

Il semble que chaque terme de cette nouvelle suite *Somme* soit un carré parfait. Vérifions en extrayant la racine carrée de chaque terme de la liste *Somme*.

```
> map(n->sqrt(n),Somme);
```

$$[1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20] \quad \textbf{(2.3)}$$

Notre intuition nous porte à croire que la somme des n premiers nombres impairs consécutifs donnera toujours le carré du nombre de termes additionnés. Autrement dit

$$\sum_{k=1}^{n} (2k-1) = n^2, \quad \forall n \in \mathbb{N}, n \geq 1$$

La formule est clairement vraie pour les 20 premiers nombres impairs (on l'a calculé). Faisons un autre essai. Additionnons, par exemple, les 39 premiers nombres impairs.

```
> add(eval(Impair,k=k),k=1..39);
```

$$1521 \tag{2.4}$$

Et

```
> sqrt(1521);
```

$$39 \tag{2.5}$$

Cela montre que la somme des 39 premiers nombres impairs consécutifs est encore un carré parfait. Cela ne contredit pas notre intuition mais cette vérification de la formule que notre intuition a permis de poser **ne prouve pas** la véracité de cette formule pour n'importe quel entier $n \geq 1$. Même si on faisait d'autres essais, cela montrerait seulement la véracité de la formule pour les différents entiers n qui auront été mis à l'essai. En fait, qu'en est-il de la somme des 21 premiers nombres impairs consécutifs ? Aussi, qu'en est-il de la somme des 22 premiers ? Etc.

***Notre intuition nous porte à croire que la somme des n premiers nombres impairs consécutifs est toujours égale au carré de n*. Mais il faut le prouver**.

Second exemple :

Soit la formule $n^2 - n + 41$. Nommons cette formule *Formule* et évaluons-la avec les entiers $n = 1, 2, 3, 4, 5,$ et 6

```
> Formule:=n^2-n+41;
```

$$Formule := n^2 - n + 41 \tag{2.6}$$

```
> for k in [1,2,3,4,5,6] do
   'Formule'[k]=eval(Formule,n=k)
  od;
```

$$Formule_1 = 41$$

$$Formule_2 = 43$$

$$Formule_3 = 47$$

$$Formule_4 = 53$$

$$Formule_5 = 61$$

$$Formule_6 = 71 \tag{2.7}$$

```
> k:='k':
```

Chacun de ces six résultats sont des nombres premiers. Faisons une autre évaluation de *Formule* avec $n = 15$.

```
> 'Formule'[15]=eval(Formule,n=15);
```

$$Formule_{15} = 251 \tag{2.8}$$

Vérifions si 251 est effectivement un nombre premier.

```
> isprime(251);
```

$$true \tag{2.9}$$

Il semble donc que l'évaluation du polynôme $n^2 - n + 41$ avec n'importe quel entier $n \geq 1$ donnera toujours un nombre premier.

Vérifions si *Formule* avec $n = 40$ est un nombre premier.

```
> isprime(eval(Formule,n=40));
```

$$true \tag{2.10}$$

Notre intuition nous porte à croire que l'évaluation de $\boldsymbol{n^2 - n + 41}$ ***donnera toujours un nombre premier quelque soit le nombre entier*** $\boldsymbol{n \geq 1}$**.** Mais il faut le prouver.

Mais, il est facile ici de voir que le calcul de $n^2 - n + 41$ ne donnera pas toujours un nombre premier. En effet, en prenant $n = 41$, on obtient 41^2 et donc $Nombre_{41}$ n'est pas un nombre premier.

```
> isprime(eval(Formule,n=41));
```

$$false \tag{2.11}$$

Ce second exemple montre bien que notre intuition peut nous induire en erreur et montre la nécessité de faire une preuve.

Raisonnement par induction mathématique

Dans la démarche scientifique, le chercheur, après avoir recueilli des observations quant à leur régularité, tire des conclusions en émettant des lois. Cette démarche est appelé démarche inductive. Par la suite, la tâche du chercheur est de "prouver" ses lois en les "vérifiant" avec des situations qui mettent à l' "épreuve" la justesse de sa théorie.

Par contre, en mathématique, il ne s'agit pas de vérifier l'universalité d'une formule avec des cas particuliers mais de faire la preuve de son universalité. Ce n'est pas en vérifiant, par exemple, quelques cas de sommes des n premiers nombres impairs consécutifs (avec $n \geq 1$) qui va établir l'universalité de la formule du résultat: n^2. Nous devons conclure, tout au plus, logiquement, que cela "marche" avec ces valeurs particulières.

Il est facile de prouver qu'une formule "ne marche pas toujours". Il suffit de montrer un cas où elle n'est pas vérifiée (contre-exemple). Mais de prouver qu'elle "marche toujours" quelque soit le cas particulier, c'est moins évident.

Lorsqu'il s'agit de faire la preuve d'énoncés mathématiques concernant les entiers positifs, le principe de l'induction mathématique est tout à fait désigné pour en faire la preuve.

Principe de l'induction mathématique

(Axiome d'induction dû à Blaise Pascal (1623-1662))

Soit $P(n)$ un prédicat défini sur les entiers et soit n_0 un certain entier.

Si on montre que

1. **P(n_0) est vrai pour un certain entier n_0,**
2. **P(k) $\Rightarrow$ P(k+1) pour tout entier $k \geq n_0$;**

Alors P(n) est vrai pour tout $n \geq n_0$.

Il y a cinq étapes à réaliser pour faire un raisonnement par induction mathématique.

Étape 1: Établir le prédicat à propos de l'entier n qui est à prouver.

Ne pas oublier de spécifier le domaine de l'entier *n*.

Étape 2: Vérification de la première condition de l'axiome d'induction. Il s'agit tout simplement de montrer que le prédicat est vrai avec $n = n_0$, c'est-à-dire que $P(n_0)$ est vraie.

Étape 3: Formuler l'hypothèse d'induction (on dit aussi hypothèse de récurrence): Il s'agit de supposer que le prédicat P(*n*) est vrai pour tout entier $n_0 < n \le k$.

Étape 4: Vérification de la seconde condition de l'axiome d'induction. Il s'agit de montrer que **si** le prédicat $P(n)$ est vrai pour tout entier *n* où $n_0 < n \le k$, il en est de même pour P(*k*+*1*). Autrement dit, il faut montrer l'implication P(*k*) ⇒ P(*k*+*1*).

Étape 5: Formulation de la conclusion.

Appliquons le principe d'induction à l'exemple de la somme des *n* premiers nombres impairs consécutifs. Cela va montrer le développement mathématique exigé par la preuve par induction mathématique..

ÉTAPE 1

Établir le prédicat, soit la proposition à propos de l'entier n qui sera à prouver. Ne pas oublier de spécifier le domaine de l'entier n.

Soit P(*n*): $1 + 3 + 5 + 7 + ... + (2n - 1) = \sum_{i=1}^{n} (2i - 1) = n^2$ pour $\forall n \in \mathbb{N}, n \ge 1$.

```
> Formule_somme:=n^2;
  Sum(2*i-1,i=1..n)=Formule_somme;
```

$$Formule\_somme := n^2$$

$$\sum_{i=1}^{n} (2i - 1) = n^2 \quad \textbf{(3.1)}$$

Montrons, en appliquant l'axiome d'induction, que P(*n*) : $\sum_{i=1}^{n} (2i - 1) = n^2$ est vrai pour $\forall n \in \mathbb{N}, n \ge 1$.

ÉTAPE 2

Vérification de la première condition de l'axiome d'induction.

Il s'agit tout simplement de montrer que la proposition est vraie avec $n_0 = 1$*, c'est-à-dire que* P(1) *est vrai.*

Le résultat obtenu directement de la somme du premier terme nous donne:

```
> Sum(2*i-1,i=1..1)=add(2*i-1,i=1..1);
```

$$\sum_{i=1}^{1} (2i - 1) = 1 \quad \textbf{(3.2)}$$

Le résultat obtenu avec la formule nous donne:

```
> eval(Formule_somme,n=1);
```

$$1 \quad \textbf{(3.3)}$$

Cela montre que P(1) est vrai. Alors, la première condition de l'axiome d'induction est vérifiée.

ÉTAPE 3

Formuler l'hypothèse d'induction (on dit aussi hypothèse de récurrence).

Supposons, par hypothèse d'induction, que P(*n*) est vrai quel que soit l'entier *n* où $1 < n \le k$, c'est-a-dire que

l'égalité

$\sum_{i=1}^{n} (2i-1) = n^2$ est vraie pour tout entier n où $1 < n \leq k$,

Par hypothèse d'induction, on pose que l'égalité $\sum_{i=1}^{k} (2i-1) = k^2$ est vraie.

```
> Hypothèse:=Sum(2*i-1,i=1..k)=eval(Formule_somme,n=k);
```

$$Hypothèse := \sum_{i=1}^{k} (2i-1) = k^2 \tag{3.4}$$

ÉTAPE 4

Vérification de la seconde condition de l'axiome d'induction. Il s'agit de montrer que si la formule $P(n)$ est vraie pour tou

Déduisons la véracité de P(k+1) avec l'hypothèse que P(k) est vrai.

La somme des (k+1) premiers nombres impairs se décompose comme la somme des k premiers nombres impairs plus le (k+1)-ième.
Le (k+1)-ième nombre impair est correspond à l'expression $2(k+1)-1$.

```
> Sum(2*i-1,i=1..k+1)=Sum(2*i-1,i=1..k)+2*(k+1)-1;
```

$$\sum_{i=1}^{k+1} (2i-1) = \sum_{i=1}^{k} (2i-1) + 2k+1 \tag{3.5}$$

Notons, dans le membre de droite, la simplification automatique $2(k+1)+1$.

Par hypothèse d'induction, $\sum_{i=1}^{k} (2i-1) = k^2$. Remplaçons dans l'égalité précédente cette sommation dans le membre de droite par k^2 .

```
> subs(Sum(2*i-1,i = 1 .. k)=k^2,(3.5));
```

$$\sum_{i=1}^{k+1} (2i-1) = k^2 + 2k+1 \tag{3.6}$$

On reconnaît dans le membre de droite le développement du carré parfait de $(k+1)$. Factorisons-le

```
> ``=factor(rhs((3.6)));
```

$$= (k+1)^2 \tag{3.7}$$

Ce résultat correspond exactement à P(n) lorsque $n = k+1$.

```
> subs(k=k+1,Hypothèse);
```

$$\sum_{i=1}^{k+1} (2i-1) = (k+1)^2 \tag{3.8}$$

Ce qui montre que la deuxième condition du principe d'induction mathématique est satisfaite: si la formule est valable avec $n = k$, elle l'est nécessairement avec $n = k+1$. Autrement dit, nous avons montré l'implication P(k) $\Rightarrow$ P($k+1$).

ÉTAPE 5

Formulation de la conclusion.

Puisque les deux conditions de l'axiome d'induction mathématique sont satisfaites, on peut donc conclure que P(n) est vrai pour tout entier $n \geq 1$, i.e.

$$\sum_{k=1}^{n} (2k-1) = n^2, \quad \forall\ n \in \mathbb{N}, n \geq 1$$

Remarque: Bien sûr, Maple aurait pu directement donner cette somme indéfinie mais **on vient de prouver que Maple a raison** de faire cette simplification automatique.

```
> Sum(2*i-1,i=1..n)=sum(2*i-1,i=1..n);
```

$$\sum_{i=1}^{n} (2i-1) = (n+1)^2 - 2n - 1 \qquad \textbf{(3.9)}$$

```
> ``=expand(rhs((3.9)));
```

$$= n^2 \qquad \textbf{(3.10)}$$

Trois autres exemples

Dans les trois autres exemples ci-dessous, nous allons appliquer le principe d'induction mathématque en respectant les cinq étapes mais sans les souligner expressément.

Exemple 1

Soit P(n): $\sum_{i=0}^{n} x^i = x^0 + x + x^2 + \ldots + x^n = \dfrac{x^{n+1}-1}{x-1}$ pour tout entier $n \geq 1$.

```
> Formule_somme:=(x^(n+1)-1)/(x-1);
  Sum(x^i,i = 0 .. n)=Formule_somme;
```

$$Formule\_somme := \frac{x^{n+1}-1}{x-1}$$

$$\sum_{i=0}^{n} x^i = \frac{x^{n+1}-1}{x-1} \qquad \textbf{(4.1.1)}$$

Faisons une preuve par induction mathématique.
Montrons que la première condition de l'axiome d'induction est satisfaite, montrons que P(n):

$\sum_{i=0}^{n} x^i = \dfrac{x^{n+1}-1}{x-1}$ avec $n_0 = 1$ est vraie.

Le résultat de la somme du premier terme nous donne:

```
> Sum(x^i,i = 0 .. 1)=sum(x^i,i = 0 .. 1);
```

$$\sum_{i=0}^{1} x^i = 1 + x \qquad \textbf{(4.1.2)}$$

Le calcul effectué à partir de la formule nous donne:

```
> Sum(x^i,i = 0 .. 1)=eval(Formule_somme,n=1);
  ``=normal(rhs(%));
```

$$\sum_{i=0}^{1} x^i = \frac{x^2-1}{x-1}$$

$$= 1 + x \quad \textbf{(4.1.3)}$$

Cela montre que P(1) est vrai. La première condition de l'axiome d'induction est satisfaite.

Montrons maintenant que la seconde condition de l'axiome d'induction est satisfaite.

Supposons, par hypothèse d'induction, que la formule *Formule_somme* est vraie pour tout entier n où $1 < n \leq k$, c'est-à-dire supposons que P(n) soit vrai jusqu'à l'entier $n = k$.

Alors, par hypothèse d'induction, nous posons l'égalité

```
> Hypothèse:=Sum(x^i,i = 0 .. k)=eval(Formule_somme,n=k);
```

$$Hypothèse := \sum_{i=0}^{k} x^i = \frac{x^{k+1} - 1}{x - 1} \quad \textbf{(4.1.4)}$$

Sous cette hypoyhèse, déduisons la véracité de P($k + 1$).

La somme des ($k + 1$) premiers termes est la somme des k premiers plus le ($k + 1$)-ième termes.

```
> Sum(x^i,i = 0 .. k+1)=Sum(x^i,i = 0 .. k)+x^(k+1);
```

$$\sum_{i=0}^{k+1} x^i = \left(\sum_{i=0}^{k} x^i\right) + x^{k+1} \quad \textbf{(4.1.5)}$$

Par hypothèse d'induction, $\sum_{i=1}^{k} x^i = \frac{x^{k+1} - 1}{x - 1}$. Remplaçons, dans l'égalité précédente, la sommation dans le membre de droite par $\frac{x^{k+1} - 1}{x - 1}$.

```
> subs(Sum(x^i,i = 0 .. k)=(x^(k+1)-1)/(x-1),%);
```

$$\sum_{i=0}^{k+1} x^i = \frac{x^{k+1} - 1}{x - 1} + x^{k+1} \quad \textbf{(4.1.6)}$$

Simplifions le membre de droite

```
> ``=simplify(rhs((4.1.6)));
```

$$= \frac{x^{k+2} - 1}{x - 1} \quad \textbf{(4.1.7)}$$

Ce résultat correspond exactement à celui qu'on obtient de la formule posée par hypothèse d'induction $\sum_{i=0}^{k} x^i = \frac{x^{k+1} - 1}{x - 1}$ en y remplaçant k par $k + 1$

```
> subs(k=k+1,Hypothèse);
```

$$\sum_{i=0}^{k+1} x^i = \frac{x^{k+2} - 1}{x - 1} \quad \textbf{(4.1.8)}$$

On vient donc de montrer que P(k) $\Rightarrow$ P($k + 1$). La seconde condition de l'axiome d'induction est satisfaite.

Les deux conditions de l'axiome d'induction mathématique étant satisfaites, on peut donc conclure que P(n) est vrai pour tout entier $n \geq 1$, i.e.

$$\sum_{i=0}^{n} x^i = \frac{x^{n+1}-1}{x-1} \quad \forall\, n \in \mathbb{N}, n \geq 1$$

Exemple 2

Soit P(n): $2\,n+1 < 2^n$, pour tout entier $n \geq 3$.

Donnons le nom P à l'inégalité $2\,n+1 < 2^n$.

```
> P:=n->2*n+1 < 2^n;
```

$$P := n \mapsto 2 \cdot n + 1 < 2^n \tag{4.2.1}$$

Faisons une preuve par induction mathématique.

Montrons que P(n): $2\,n+1 < 2^n$ avec $n_0 = 3$ est vrai.

Obtenons directement la valeur de vérité de P(n) avec $n_0 = 3$.

```
> evalb(P(3));
```

$$\mathit{true} \tag{4.2.2}$$

Cela montre que P(3) est vrai, la première condition de l'axiome d'induction est satisfaite.

Supposons, par hypothèse d'induction, que P(n) est vrai pour tout entier n où $3 \leq n \leq k$, c'est-à-dire supposons que P(n) est vrai jusqu'à l'entier $n = k$ inclusivement.

Alors, par hypothèse d'induction, nous avons.

```
> Hypothèse:=P(k);
```

$$\mathit{Hypothèse} := 2\,k+1 < 2^k \tag{4.2.3}$$

Additionnons 2 à chaque membre de l'inégalité. L'inégalité obtenue est encore vraie évidemment.

```
> P(k)+(2=2);
```

$$2\,k < -1 + 2^k \tag{4.2.4}$$

Cette simplification automatique n'est pas souhaitée ici. Pour qu'on puisse déduire clairement que $2\,n+1 < 2^n$, c'est-à-dire déduire que P($k+1$) est vraie, il nous faut contourner le mécanisme de la simplification de Maple.

```
> P(k)+(deux=deux);
```

$$\mathit{deux} + 2\,k + 1 < \mathit{deux} + 2^k \tag{4.2.5}$$

Puisque la plus petite valeur de 2^k est 8 ($k = 3$), a fortiori, l'inégalité $2^k + 2 < 2\,2^k$ est vraie. D'où, par transitivité, la véracité de l'inégalité suivante:

```
> subs(deux=2,lhs((4.2.5)))<rhs(P(k))*2;
```

$$3 + 2\,k < 2\,2^k \tag{4.2.6}$$

```
> simplify((4.2.6));
```

$$3 + 2\,k < 2^{k+1} \tag{4.2.7}$$

Ce résultat correspond exactement à celui qu'on obtient de la formule posée par hypothèse d'induction $2\,k+1 < 2^k$ en y remplaçant k par $k+1$.

```
> subs(k=k+1,Hypothèse);
```

$$3 + 2\,k < 2^{k+1} \tag{4.2.8}$$

On vient donc de montrer que $\mathrm{P}(k) \Rightarrow \mathrm{P}(k+1)$.

Puisque les deux conditions de l'axiome d'induction mathématique sont satisfaites, on peut donc conclure que $\mathrm{P}(n)$ est vrai pour tout entier $n \geq 3$, i.e.

$$2n+1 < 2^n,\ \forall\ n \in \mathbb{N}, n \geq 3.$$

Exemple 3

Soit $\mathrm{P}(n)$: 3 divise ($2^{2n}-1$), c'est-à-dire que $3 \mid (2^{2n}-1)$ pour tout entier $n \geq 1$.

Donnons le nom *Nombre* à l'expression $2^{2n}-1$.

```
> Nombre:=2^(2*n)-1;
```

$$Nombre := 2^{2n}-1 \quad \textbf{(4.3.1)}$$

Faisons une preuve par induction mathématique.

Montrons que la première condition de l'axiome d'induction est satisfaite. Montrons que $3 \mid 2^{2n}-1$ avec $n_0 = 1$ est vrai. Évaluons $\mathrm{P}(n)$ avec $n_0 = 1$.

```
> eval(Nombre,n=1);
```

$$3 \quad \textbf{(4.3.2)}$$

Le résultat est clairement divisible par 3. Cela montre que P(1) est vrai. La première condition de l'axiome d'induction est satisfaite.

Montrons maintenant que la seconde condition de l'axiome d'induction est satisfaite.

Supposons, par hypothèse d'induction, que $2^{2n}-1$ est divisible par 3 jusqu'à l'entier $n = k$ inclusivement, autrement dit que $2^{2n}-1$ est un multiple de 3 jusqu'à l'entier k inclusivement.

Alors, par hypothèse d'induction, on pose, pour un certain entier q, $2^{2k} = 3q$.

```
> Hypothèse:=2^(2*k)-1=3*q;
```

$$Hypothèse := 2^{2k}-1 = 3q \quad \textbf{(4.3.3)}$$

Sous cette hypothèse, déduisons la véracité de $\mathrm{P}(k+1)$.

Remplaçons n dans la formule $2^{2n}-1$ par $k+1$.

```
> 'P(k+1)'=subs(n=k+1,Nombre);
```

$$P(k+1) = 2^{2k+2}-1 \quad \textbf{(4.3.4)}$$

Développons.

```
> ``=combine(rhs((4.3.4)),power);
```

$$= 4\,2^{2k}-1 \quad \textbf{(4.3.5)}$$

Par l'hypothèse d'induction, nous avons que $2^{2k}-1 = 3q$, pour un certain entier q. Remplaçons donc 2^{2k} par $3q+1$ dans cette dernière expression.

```
> ``=subs(2^(2*k)=3*q+1,rhs((4.3.5)));
```

$$= 12q+3 \quad \textbf{(4.3.6)}$$

Divisons ensuite ce résultat $12q+3$ par 3.

```
> ``=rhs((4.3.6))/3;
```

$$= 4\,q + 1 \tag{4.3.7}$$

Puisque q est un entier, $4\,q + 1$ en est un aussi, cela montre que 3 divise $2^{2\,k + 2} - 1$.

On vient donc de montrer que P(k) ⇒ P($k + 1$).

Puisque les deux conditions de l'axiome d'induction mathématique sont satisfaites, on peut donc conclure que P(n) est vrai pour tout entier $n \geq 1$, 3 divise $(2^{2\,n} - 1)$, c'est-à-dire que

$$3 \mid (2^{2\,n} - 1) \quad \forall\, n \in \mathbb{N}, n \geq 1\,.$$

Exercices

Appliquer l'axiome d'induction pour prouver les formules ci-dessous. Rappelons cet axiome.

Soit $P(n)$ un prédicat défini sur les entiers et soit n_0 un certain entier.

Si on montre que

1. P(n_0) est vrai pour un certain entier n_0,

2. P(k) ⇒ P(k+1) pour tout entier $k \geq n_0$;

Alors P(n) est vrai pour tout $n \geq n_0$.

En utilisant l'axiome d'induction, montrer les prédicats P(n) suivantes.

No.1 P(n): $1 + 2 + 3 + 4 + 5 + ... + \mathrm{n} = \dfrac{n\,(n+1)}{2}$, $\forall\, n \in \mathbb{N}, n \geq 1$.

No.2 P(n): $1 + 2 + 4 + 8 + ... + 2^{n-1} = 2^n - 1$, $\forall\, n \in \mathbb{N}, n \geq 1$.

No.3 P(n): $a + ar + ar^2 + ... + ar^{n-1} = \dfrac{a\left(1 - r^n\right)}{1 - r}$, $\forall\, n \in \mathbb{N}, n \geq 1$ et $r \neq 1$.

No.4 P(n): $\displaystyle\sum_{i=0}^{n} r^i = \frac{r^{n+1} - 1}{r - 1}$, $\forall\, n \in \mathbb{N}$ et $r \neq 1$.

No.5 P(n): $\displaystyle\sum_{k=1}^{n} \frac{1}{k\,(k+1)} = \frac{n}{n+1}$, $\forall\, n \in \mathbb{N}, n \geq 1$

No.6 P(n): $n^2 + n$ est divisible par 2, $\forall\, n \in \mathbb{N}, n \geq 1$

No.7 P(n): $n^3 + 2\,n$ est divisible par 3, $\forall\, n \in \mathbb{N}, n \geq 1$

No.8 P(n): $5^n - 1$ est divisible par 4, $\forall\, n \in \mathbb{N}, n \geq 1$

No.9 P(n): $n\left(n^2 + 5\right)$ est divisible par 6, $\forall\, n \in \mathbb{N}, n \geq 1$

No.10 P(n): $\dfrac{1}{1.4} + \dfrac{1}{4.7} + \dfrac{1}{7.10} + ... + \dfrac{1}{(3\,n - 2)\,(3\,n + 1)} = \dfrac{n}{3\,n + 1}$, $\forall\, n \in \mathbb{N}, n \geq 1$